Visual Basic for Applications

Module 5
Introduction

The purpose of this module is to introduce VBA programming with Microsoft Access.  By the end of this module you should be able to accomplish the following tasks:

· Create a new Microsoft Access database from within Excel. 



User Forms
User forms put the Visual in Visual Basic.  A nice Windows-style form can be created quickly and easily using the forms designer and your mouse.
In this example we will create a user form in Excel that can count how many formulas are in a selection, sheet, or entire workbook.

Designing the User Form in Excel
Open Excel and enter the VBE.  Insert a User Form into your Personal.xls file.  A new blank user form is created.  Make sure the Toolbox and the Properties are showing by using the View menu. 

Now right-click on the Toolbox and choose Additional Controls.  Make sure you have at least the following items checked in your list

Using the toolbox design a form that looks like the following picture.  You will also need to use the Properties window to set certain properties.

The following table lists the names you should give to each element.
	Element
	(Name) in Properties Window

	This Sheet Option Button
	optThis

	Entire Workbook Option Button
	optEntire

	Selection Option Button
	optSelection

	Ok Button
	cmdOk

	Close Button
	cmdClose

	Frame around Reference Edit control
	Frame2


Adding Code to the User Form

Simple Design Functionality
Now that we have designed the form we need to add code to give it functionality.  To quickly add code to an item, double-click it and you will be at the default event for an item.  For example, double-click on the Close button and it will open to the code editor in the cmdClose_Click event.  Type the following lines of code in the Click event.
1
Unload Me

2
End

Line 1 causes the form to be unloaded.  This will close the form and remove it from memory.

Line 2 forces the code to end.  This isn’t necessary in this example because we only have one form so when we unload the one form it will automatically end.

To test your code switch back to design view of the user form and click Run, Run Sub/UserForm.  Click the Close button to make sure it closes the form.

Now we want to add code so that the bottom Reference Edit control is only viewable when we select the Selection option button.  First double-click on the This Sheet option button.  This will open the code editor in the optThis_Click() event.  Add the following line of code.

Frame2.Visible = False
This will hide the Reference Edit control.  Now do the same with the Entire Workbook option button.  Add code to make Frame2 visible when optSelection is clicked.  Also add code to Set Focus to the Frame.  We also want Selection to be selected automatically every time we run the form.  From the Code Editor choose UserForm from the General drop-down list and Initialize from the Declarations drop-down list.  Add this line of code.

optThis.Value = True

Test the form again to make sure that the Reference Edit control is only viewable when we select Selection.

Main Code of the Form

We now have the simple functionality working.  We now need to add code to count the formulas.  Open the Click event for the cmdOk button and type the following code.
1
Dim Total As Double, SubTotal As Double

2
Dim i As Integer

3
Total = 0

4
5
If optThis.Value = True Then

6
    On Error Resume Next

7
    Total = Selection.SpecialCells(xlCellTypeFormulas, 23).Count

8
End If

9
If optEntire.Value = True Then

10
    For i = 1 To ActiveWorkbook.Sheets.Count

11
        SubTotal = 0

12
        Sheets(i).Activate

13
        On Error Resume Next

14
        SubTotal = Selection.SpecialCells(xlCellTypeFormulas, 23).Count

15
        Total = Total + SubTotal

16
    Next i

17
End If

18
If optSelection.Value = True Then

19
    Dim SelRange As Range

20
    Dim address As String

21
    address = RefEdit1.Value

22
    Set SelRange = Range(address)

23
    On Error Resume Next

24
    Total = SelRange.SpecialCells(xlCellTypeFormulas, 23).Count

25
End If

26
MsgBox "There were " & Format(Total, "###,###,###") & " formulas counted." 

Line 3 resets our total count.
Line 5 starts the code if This Sheet was selected.

Line 6 turns off error checking because Selection.SpecialCells will return an error if there are none.

Line 9 starts the code if the Entire Workbook was selected.
Line 10 starts a loop through all of the sheets in the workbook.

Lines 11-15 activate each sheet and add the subtotals together.
Line 18 starts the code if Selection was selected.

Lines 19-22 get the range address that was selected in the user form.

Line 26 displays a Message Box with total number of formulas counted.
Exercise 1
· Add code to calculate how long it takes to count the formulas.  Declare a variable in the cmdOk_Click() event and set its value to Now().  When the Message Box is displayed at the end, add a line that tells how long it took to count the formulas.
· Include code to return us to the sheet we invoked the user form from.  For example, the code now will end on the last sheet if we choose Entire Workbook.  Store which sheet we are on when the form is first loaded.  Then return to that sheet when our code is done.

· Add code to change the forms mouse pointer to an hourglass while the formulas are being counted.

· Customize your toolbars in Excel so that you have quick access to the user form.



User Form in Microsoft Word

In the following example we will see how to create a user form in Microsoft Word that pulls data from an Access database.  (This example was inspired by the Katy account and Word template I used for Katy.)  You should have received a file in your email named Addresses.mdb.  We want to create a form that will read the data in from the Access database and allow us to choose which person’s information we want.
Designing the Form

Open Word and enter the VBE.  Insert a User Form into your Normal.dot and design it to look like the following.  ( Note: The large white box is a list box. )

	Element
	(Name) in Properties Window

	Get Names button
	cmdGetNames

	Paste and Close button
	cmdPasteandClose

	Cancel Button
	cmdCancel

	Names list
	listNames


Adding Code to the form

Add code to the Cancel button to unload the form.
We need to add some references so that we can connect to the Access database.  Click on Tools, References and choose Microsoft ActiveX Data Objects 2.7 Library (or the latest number your computer has. )

Enter the code window for the user form and scroll all the way to the top.  Add the following lines of code before any other lines already there.

1
Dim goConn As ADODB.Connection

2
Dim rsAddresses As ADODB.Recordset

3
Dim i As Integer, cnt As Integer
Line 1 declares a variable that we will use to connect to the Access database.
Line 2 declares a variable that we will use to pull the records from the database

Line 3 declares variables that will be used in loops.

Double-click the Get Names button and enter the following lines of code.

1
Set goConn = New Connection

2
goConn.CursorLocation = adUseClient

3
goConn.Open ("Provider=Microsoft.Jet.OLEDB.4.0;Data Source=g:\Addresses.mdb;Persist Security Info=False")

4

5
Set rsAddresses = New Recordset

6
rsAddresses.Open "Select * FROM tblAddresses ORDER BY FullName", goConn, adOpenKeyset, adLockOptimistic

7

8
listNames.Clear

9
If rsAddresses.RecordCount Then

10
    rsAddresses.MoveFirst

11
    cnt = rsAddresses.RecordCount

12
    For i = 1 To cnt

13
        listNames.AddItem rsAddresses!FullName
14
        rsAddresses.MoveNext

15
    Next i

16
End If
Line 1 creates a new connection in memory

Line 3 opens a connection to our database.  Modify the path to wherever you saved the addresses.mdb file.

Line 5 creates a new record set in memory.  A record set is basically the same thing as a table in Microsoft Access.

Line 6 opens the record set and fills it with everything (Select * will select everything in the table or you can specify fields, i.e. Select FirstName, LastName)

Line 9 makes sure there are records returned

Lines 12-15 loop through all of the records and line 13 adds the first and last name to our list box. 
Run the form now and make sure the Get Names button works.

Now we simply want to insert the name and address in a mail format into our document wherever the cursor is.  Open the code for the Paste and Close button and enter the following lines.

1
rsAddresses.MoveFirst

2
For i = 0 To listNames.ListCount - 1

3
    If listNames.Selected(i) Then

4
        Dim strAddress As String

5
        rsAddresses.Find "FullName = '" & listNames.List(listNames.ListIndex) & "'"

6
        strAddress = rsAddresses!FullName & vbCrLf & rsAddresses.Fields("StreetAddress").Value _

7
        & vbCrLf & rsAddresses!City & ", " & rsAddresses!State & "   " & rsAddresses!ZipCode

8
        Selection.text = strAddress

9
    End If

10
Next i

11
Unload Me
Line 1 moves to the first record in our record set.

Line 2 starts a loop through the list box to see which name is selected.

Line 3 checks to see if the name is selected.

Line 4 declares a variable that will be used to build our string containing the mailing address.

Line 5 finds the record that corresponds to the name we selected in the list box.

Lines 6 and 7 build the mailing address using the different fields in the record set.

Line 6 has a continuation mark at the end “_” which is used to continue the same line onto a new line.  In other words, without the “_” lines 6 and 7 would need to show on the same line in the VBE.

Line 8 sets the text at the current cursor location in our Word document to the string we built.

Run the form and make sure everything works.
Exercise 2

· Add a text box below the list of names that will show a preview of the addresses.
· Add code so that when you click on the list box and select a name the text box below shows a preview of the address.

· Change the Paste and Close button so that it puts the data from the text box preview into the Word document.


Summary

This module introduces VBA User Forms and shows how quickly and easily a standard Windows form can be created.  This module also introduces and touches briefly how to write code to work with an Access database.  A future module will be dedicated to a more in depth look at programming in Microsoft Access.

